

NATIONAL RADIO ASTRONOMY OBSERVATORY
Charlottesville, Virginia

Electronics Division Technical Note No. 157

Title: HP-8510 NETWORK ANALYZER DATA ACQUISITION BY FARANT

Author(s): J. Daniel Gallego

Date: December 20, 1989

DISTRIBUTION:

<u>GB</u>	<u>CV</u>	<u>TU</u>	<u>VLA</u>
GB Library	ER Library	Library Downtown	VLA Library
R. Lacasse	IR Library	Library Mountain	P. Napier
D. Schiebel	M. Balister	R. Freund	J. Campbell
E. Childers	C. Burgess	J. Lamb	W. Brundage
C. Brockway	S. Srikanth	D. Emerson	P. Lilie
J. Coe	S-K. Pan	P. Jewell	R. Weimer
R. Norrod	A. R. Kerr	J. Cochran	
S. White	N. Bailey	A. Perfetto	
G. Behrens	L. D'Addario		
R. Fisher	N. Horner		
F. Crews			

HP-8510 NETWORK ANALYZER DATA ACQUISITION BY FARANT

J. Daniel Gallego

Introduction

This note describes two additions to the "FARANT" program which facilitate the use of data measured by the HP network analyzer (HP-8510) by the "FARANT" subroutines. First is a program "FILEMGR" (for File Manager) which creates a file from the HP-8510 measured data and also allows for its editing, display, and printing. Second is a subroutine "Newfile" which can be used to read files created by "FILEMGR" for further use by "FARANT" subroutines.

Program "FILEMGR"

Program "FILEMGR" is stored on the main "FARANT" disk. Before running this program, the following preparatory steps must be taken:

- the network analyzer's HPIB address must be set at 16 (this is the factory setting)
- the HPIB interfaces of the computer and network analyzer must be connected
- an adequate calibration of the network analyzer must be performed or recalled
- the proper reference planes must be set, if different from those at which calibration was performed
- a not-write-protected disk with sufficient space to store the data must be placed in current "Mass Storage Is" device

Upon execution, program "FILEMGR" displays a six-option menu:

1) CREATE A NEW FILE

This option creates a new file from HP-8510 measured S-parameters. It asks for a frequency range (F_{\min} , F_{\max}) and a frequency step (F_{step}) for which data is to be stored. It also asks for a file name and a file header (a string up to 36 characters long). If the same file name already exists in "MASS STORAGE IS" device, the program asks for permission to overwrite it. Option #1 requires the network analyzer to be calibrated within the frequency band of interest or larger. Any value of F_{\min} , F_{\max} , and F_{step} may be chosen by the user in calibrated frequency range. Each S-parameter is measured with the average factor of 16. This should be sufficient in most practical applications. If it is not adequate, like for instance, in measurement of very small reflection coefficients, it may be changed by the user. It requires the editing of the program and changing the values AVERON (Average On) and NUMG (Number of Groups) parameters. Always

NUMG = AVERON + 1

A different value of AVERON and NUMG may be assigned in line 10650 to a given S-parameter measurement.

A measurement is done with MARKER #1 set to each specified frequency with the CONTINUOUS MARKER option ON. If that frequency is different from the frequency stored by the network analyzer in the calibration procedure, the instrument interpolates between two nearby frequency points. After the completion of measurement, the original network analyzer settings are restored.

2) DELETE OLD FILE

This option may be used to purge old data files. The program prompts for a given file name.

3) READ FILE HEADER

This option allows inspection of data file headers which contain F_{\min} , F_{\max} , F_{step} , and a comment, 36 characters long.

4) DISPLAY FILE

This option allows the display of data files at CRT.

5) PRINT FILE

This option allows printing of a data file with header by the "PRINTER IS" device (701).

6) EXIT

This option terminates the execution of the program.

The organization of a data file is as follows:

File type: BDAT

Record size: 64 bytes (8*REAL) (8 REAL Numbers)

Number of records: INT(DROUND((F_{\max} - F_{\min})/ F_{step}),12) + 2

A file can be accessed randomly. The first record contains a header with three real numbers: a frequency range (F_{\min} , F_{\max}), a step frequency (F_{step}), and also a 36-character string for commenting purposes. Each subsequent record holds the four S-parameters at each frequency point in the following order: S11, S12, S21, S22. Parameters are given in polar coordinates (MAG./ANG.(DEG.)).

Subroutine "Newfile" (X(*), Filename\$, F_{\min} , F_{\max} , F_{step})

This new FARANT subroutine, which is stored in FARAFT, can be used to read files created by "FILEMGR". The matrix X(*) must be dimensioned as X(M,6,4) where M is the number of frequency points. The maximum value of M is 51 as there

are only 51 frequency points in FARANT data base Dat(*). X(*) is, therefore, an array of 6x4 matrices, each describing a two-port at a given frequency in S-parameters representation. That is:

```
X(1,*,*) → Data for F = Fmin
X(2,*,*) → Data for F = Fmin + Fstep
•
•
•
•
X(M,*,*) → Data for F = Fmax
```

Total number of frequency points, M, is:

$$M = \text{INT}(\text{DROUND}((F_{\max} - F_{\min})/F_{\text{step}}, 12)) + 1$$

The Filename\$ is the name of the file created by "FILEMGR" from which the data are to be loaded into X(*). The frequency range (F_{\min} , F_{\max}) and frequency step (F_{step}) specified as parameters of subroutine "NEWFILE" may be different from those stored in the file Filename\$. However, each specified frequency point must be in the file. For example, if Filename\$ holds data from 1.0 to 2.0 GHz in 0.1 GHz steps, "NEWFILE" can choose data from 1.2 to 1.8 GHz in 0.2 GHz steps and store it in X(*). A warning message is displayed if a requested frequency point is not in the file.

A reading of a file Filename\$ is a time consuming process. If the file data are to be used in an optimization, the data upon reading should be stored in a labeled common-declared matrix.

The features of the HPBASIC MAT Option allow for a very simple loading of 6x4 FARANT variables from X(*) in a frequency loop, as for example:

```
FOR I = 1 to M
  F = Fmin + (I - 1) * Fstep
  •
  •
  •
  MAT A = X(I,*,*)
  •
  •
  •
  NEXT I
```

Matrix A(*), above, can be operated in the same way as for all FARANT two-ports.

Program COILFIT in the Appendix is an example of fitting parameters of the coil equivalent circuit to the measured S-parameters which are read from the file COIL.

APPENDIX

Program "FILEMGR"

Subroutine "NEWFILE"

Example: "COILFIT"

PROGRAM "FILEMGR"

```
10000 ! ***** PROGRAM FILEMGR *****
10010 OPTION BASE 1
10020 INTEGER O,Size,Preamble
10030 DIM Filename$(10),Header$(36),Test$(1),Dat(8)
10040 INTEGER Learn_string(2195)
10050 PRINTER IS CRT
10060 ASSIGN @Hplib TO 7
10070 ASSIGN @Nwa TO 716
10080 ASSIGN @Nwa_data1 TO 716;FORMAT ON
10090 ASSIGN @Nwa_data2 TO 716;FORMAT OFF
10100 !
10110 Menu:    ! SCREEN MENU ****
10120 !
10130 OUTPUT KBD;CHR$(255)&"K";   ! CLEARING THE SCREEN
10140 PRINT TABXY(30,2);"PROGRAM FILEMGR"
10150 PRINT TABXY(1,4);"This program reads 'S' parameters from the HP-8510 Network Analyzer."
10160 PRINT TABXY(1,6);"You must set in the 8510 an adequate frequency range and a calibration before attempting to load the file."
10170 PRINT TABXY(1,9);"This program re-stores your previous configuration to the NWA after reading."
10180 PRINT TABXY(27,13);"OPTIONS:"
10190 PRINT TABXY(9,14);"-----"
10200 PRINT TABXY(10,15);"1-CREATE NEW FILE"
10210 PRINT TABXY(10,16);"2-DELETE OLD FILE"
10220 PRINT TABXY(10,17);"3-READ FILE HEADER"
10230 PRINT TABXY(35,15);"4-DISPLAY FILE (CRT)"
10240 PRINT TABXY(35,16);"5-PRINT FILE (PRINTER)"
10250 PRINT TABXY(35,17);"6-EXIT"
10260 INPUT "OPTION?",O
10270 ON O GOSUB Create,Delete,Header,Display,Print,Exit
10280 GOTO Menu
10290 !
10300 Create: ! LOAD A NEW FILE ****
10310 !
10320 OUTPUT KBD;CHR$(255)&"K";
10330 PRINT TABXY(1,10);"Enter the minimum, maximum, frequency step."
10340 INPUT "Minimum, Maximum, Step?",Fmin,Fmax,Fstep
10350 PRINT TABXY(1,10);"Enter the name for the file (max. 10 chr.) "
10360 LINPUT "File name?",Filename$
10370 IF Filename$="" THEN GOTO 10360
10380 PRINT TABXY(1,10);"Enter the header label (max. 36 chr.)      "
10390 LINPUT "Header label?",Header$
10400 OUTPUT KBD;CHR$(255)&"K";
10410 PRINT "File: ";Filename$
10420 PRINT "Header: ";Header$
10430 PRINT "Fmin= ";Fmin
10440 PRINT "Fmax= ";Fmax
10450 PRINT "Step= ";Fstep
10460 Records=INT(DROUND((Fmax-Fmin)/Fstep,12))+2
10470 ON ERROR GOSUB Recov
10480 CREATE BDAT Filename$,Records,8*8
10490 OFF ERROR
10500 ALLOCATE Dat1(Records-1,8)
10510 ASSIGN @File TO Filename$
10520 OUTPUT @File,1;Fmin,Fmax,Fstep,Header$
10530 !
10540 ! ***** STORE NWA CONFIGURATION *****
10550 !
10560 ABORT @Hplib
```

```

10570 REMOTE @Hplib
10580 OUTPUT @Nwa;"OUTPLEAS;"
10590 ENTER @Nwa_data2;Preamble,Size,Learn_string(*)
10600 !
10610 !***** MEASURING "S" PARAMETERS *****
10620 !
10630 DISP "MEASURING 'S' PARAMETERS....."
10640 OUTPUT @Nwa;"DELO;CHAN1;LINP;MARKCONT;SINC;"
10650 OUTPUT @Nwa;"HOLD;AVERON 16;NUMG 17;"
10660 OUTPUT @Nwa;"MARK1;"
10670 !
10680 ! ***** READ S11 AND STORE IN MAT DAT1 *****
10690 !
10700 OUTPUT @Nwa;"S11;LINP;WAIT;"
10710 DISP "READING S11"
10720 FOR I=2 TO Records
10730   F=Fmin+(I-2)*Fstep
10740   OUTPUT @Nwa;F;"GHZ;OUTPMARK;"
10750   ENTER @Nwa_data1;Dat1(I-1,1),Dat1(I-1,2)
10760 NEXT I
10770 !
10780 ! ***** READ S12 AND STORE IN MAT DAT1 *****
10790 !
10800 OUTPUT @Nwa;"S12;LINP;WAIT;"
10810 DISP "READING S12"
10820 FOR I=2 TO Records
10830   F=Fmin+(I-2)*Fstep
10840   OUTPUT @Nwa;F;"GHZ;OUTPMARK;"
10850   ENTER @Nwa_data1;Dat1(I-1,3),Dat1(I-1,4)
10860 NEXT I
10870 !
10880 ! ***** READ S21 AND STORE IN MAT DAT1 *****
10890 !
10900 OUTPUT @Nwa;"S21;LINP;WAIT;"
10910 DISP "READING S21"
10920 FOR I=2 TO Records
10930   F=Fmin+(I-2)*Fstep
10940   OUTPUT @Nwa;F;"GHZ;OUTPMARK;"
10950   ENTER @Nwa_data1;Dat1(I-1,5),Dat1(I-1,6)
10960 NEXT I
10970 !
10980 ! ***** READ S22 AND STORE IN MAT DAT *****
10990 !
11000 OUTPUT @Nwa;"S22;LINP;WAIT;"
11010 DISP "READING S22"
11020 FOR I=2 TO Records
11030   F=Fmin+(I-2)*Fstep
11040   OUTPUT @Nwa;F;"GHZ;OUTPMARK;"
11050   ENTER @Nwa_data1;Dat1(I-1,7),Dat1(I-1,8)
11060 NEXT I
11070 !
11080 ! ***** RESTORE NWA STATUS *****
11090 !
11100 OUTPUT @Nwa;"INPULEAS;"
11110 OUTPUT @Nwa_data2;Preamble,Size,Learn_string(*)
11120 LOCAL @Hplib
11130 !
11140 ! ***** LOADS FILE WITH DAT *****
11150 !
11160 DISP "LOADING FILE "";Filename$;"""

```

```
11170 FOR I=2 TO Records
11180   MAT Dat= Dat1(I-1,*)
11190   OUTPUT @File,I;Dat(*)
11200 NEXT I
11210 DEALLOCATE Dat1(*)
11220 ASSIGN @File TO *
11230 RETURN
11240 !
11250 Delete: !DELETE FILE SELECTED
11260 !
11270 OUTPUT KBD;CHR$(255)&"K";
11280 LINPUT "Name of the file to be DELETED?",Filename$
11290 PURGE Filename$
11300 RETURN
11310 !
11320 Header: !READS THE HEADER OF A FILE
11330 !
11340 OUTPUT KBD;CHR$(255)&"K";
11350 LINPUT "Name of the file to be READ?",Filename$
11360 ASSIGN @File TO Filename$
11370 ENTER @File,1;Fmin,Fmax,Fstep,Header$
11380 PRINT "File: ";Filename$
11390 PRINT "Header: ";Header$
11400 PRINT "Fmin= ";Fmin
11410 PRINT "Fmax= ";Fmax
11420 PRINT "Step= ";Fstep
11430 ASSIGN @File TO *
11440 LINPUT "",Test$
11450 RETURN
11460 !
11470 Display: ! *****
11480 !
11490 OUTPUT KBD;CHR$(255)&"K";
11500 LINPUT "Name of the file to be DISPLAYED?",Filename$
11510 GOSUB Out
11520 LINPUT "",Test$
11530 RETURN
11540 !
11550 Print: ! *****
11560 !
11570 LINPUT "Name of the file to be PRINTED?",Filename$
11580 PRINTER IS 701
11590 GOSUB Out
11600 PRINTER IS CRT
11610 RETURN
11620 !
11630 Out: ! *****
11640 !
11650 ASSIGN @File TO Filename$
11660 ENTER @File,1;Fmin,Fmax,Fstep,Header$
11670 PRINT "File: ";Filename$
11680 PRINT "Header: ";Header$
11690 PRINT "Fmin= ";Fmin
11700 PRINT "Fmax= ";Fmax
11710 PRINT "Step= ";Fstep
11720 Records=INT(DROUND((Fmax-Fmin)/Fstep,12))+2
11730 PRINT
11740 PRINT USING 11750
11750 IMAGE 14X,"$11",13X,"$12",13X,"$21",13X,"$22"
11760 PRINT USING 11770
```

```
11770 IMAGE " FREQ",4(5X,"MAG",5X,"ANG")
11780 PRINT USING 11790
11790 IMAGE "-----",4(2X,"-----",X,"-----")
11800 FOR I=2 TO Records
11810   F=Fmin+(I-2)*Fstep
11820   ENTER @File,I;Dat(*)
11830   PRINT USING 11840;F,Dat(*)
11840   IMAGE DD.DDD,2X,4(DDD.DDD,X,DDDD.D,2X)
11850 NEXT I
11860 ASSIGN @File TO *
11870 RETURN
11880 !
11890 Recov!:! *** RECOVERS FROM DUPLICATE FILE NAME ERROR ***
11900 !
11910 IF ERRN=54 THEN
11920   PRINT TABXY(1,10);"Duplicate file '';Filename$;'' exist on disc. Do you
want to OVERWRITE? (Y/N)"
11930   LINPUT "Y/N?",Test$
11940   PRINT TABXY(1,10);"
"
11950   IF Test$="Y" OR Test$="y" THEN
11960     PURGE Filename$
11970   ELSE
11980     LINPUT "New file name?",Filename$
11990     PRINT TABXY(1,1), "File:  ";Filename$;""
12000 END IF
12010 ELSE
12020   OFF ERROR
12030 END IF
12040 RETURN
12050 !
12060 Exit!:! ****
12070 !
12080 OUTPUT KBD;CHR$(255)&"K";
12090 STOP
12100 END
```

SUBROUTINE "NEWFILE"

```
10000 SUB Newfile(X(*),Filename$,Fmin,Fmax,Fstep) !*****
10005 !
10010 ! THIS SUB READS 'S' PARAMETERS IN FILE Filename$, AND STORES THE
10015 ! RESULTS IN MATRIX X(*), WHITCH MUST BE DIMENSIONED IN SUB Cktana
10020 !
10025 OPTION BASE 1
10030 INTEGER I,N,M,Nini,Nend,Nstep
10035 DIM R(8)
10040 ASSIGN @File TO Filename$
10045 DEG
10050 MAT X= (0) ! CLEANS OLD NOISE
10055 ENTER @File,1;Fminf,Fmaxf,Fstepf
10060 IF Fmin<Fminf OR Fmax>Fmaxf OR Fstep<Fstepf THEN GOTO Warning
10065 IF FRACT(DROUND((Fstep/Fstepf),12))>1.E-9 THEN GOTO Warning
10070 IF FRACT(DROUND((Fmin-Fminf)/Fstepf,12))>1.E-9 THEN GOTO Warning
10075 Nini=INT(DROUND((Fmin-Fminf)/Fstepf,12))+1
10080 Nend=INT(DROUND((Fmax-Fminf)/Fstepf,12))+1
10085 Nstep=INT(DROUND((Fstep/Fstepf),12))
10090 I=0
10095 FOR N=Nini TO Nend STEP Nstep
10100 I=I+1
10105 F=Fminf+(N-1)*Fstepf
10110 DISP "READING FILE AT F=";F;" GHz"
10115 ENTER @File,N+1;R(*)
10120 !
10125 ! CHANGE TO RE,IMAG
10130 !
10135 FOR M=1 TO 7 STEP 2
10140 Mag=R(M)
10145 Phase=R(M+1)
10150 R(M)=Mag*COS(Phase)
10155 R(M+1)=Mag*SIN(Phase)
10160 NEXT M
10165 !
10170 ! LOAD X(I,*) WITH 'S' PARAMETERS (TIPE 4)
10175 !
10180 X(I,5,1)=4 !Pset
10185 X(I,1,1)=R(1)
10190 X(I,2,2)=R(1)
10195 X(I,1,2)=R(2)
10200 X(I,2,1)=-R(2)
10205 X(I,1,3)=R(3)
10210 X(I,2,4)=R(3)
10215 X(I,1,4)=R(4)
10220 X(I,2,3)=-R(4)
10225 X(I,3,1)=R(5)
10230 X(I,4,2)=R(5)
10235 X(I,3,2)=R(6)
10240 X(I,4,1)=-R(6)
10245 X(I,3,3)=R(7)
10250 X(I,4,4)=R(7)
10255 X(I,1,4)=R(8)
10260 X(I,4,3)=-R(8)
10265 NEXT N
10270 DISP
10275 SUBEXIT
10280 Warning: !
10285 PRINT "FREQUENCY DATA IN FILE '">Filename$;"' DOESN'T FIT"
10290 STOP
10295 SUBEND
```

EXAMPLE: "COILFIT"

```
10000 SUB Farstart           !USER'S CONTROL OF FARANT BEGINS HERE #####
10005   PRINTER IS 701
10010   OPTION BASE 1
10015   INTEGER N
10020   READ N               !# OF PARAMETERS TO OPTIMIZE
10025   ALLOCATE X(N)
10030   READ X(*)             !FOR INITIAL GUESSES ONLY WHEN OPTIMIZING
10035   DATA 3,1,1,1          !PUT N, INITIAL GUESSES HERE (USE NO ZEROS)
10040   Cktanalysis(X(*),0,2) !FOR PRE-OPTIMIZED ANALYSIS; THE DEFAULT
10045   Optimize(X(*),1)     !MAKE THIS A STATEMENT TO DO OPTIMIZATION
10050 SUBEND
10055 SUB Cktanalysis(X(*),Fvalue,INTEGER Opt)           ######
10060 ! **** VERSION OF COILFIT ***
10065! WHEN Opt=1 ASSIGN Fvalue; OTHERWISE DO NORMAL ANALYSIS & OUTPUT
10070   OPTION BASE 1
10075   COM Zo,F,Dat(*),INTEGER Nogo,Count  ![Dat] HOLDS FREQ, CKT & NOISE
10080   COM /Ckt/ Filedata(51,6,4),Fmin,Fmax,Fstep,INTEGER M ! HOLDS FILE DATA
10085   DIM A(6,4),B(6,4),C(6,4),D(6,4),E(6,4)
10090   INTEGER I,J,K,L,N
10095   Count=0                !Count = #FREQS CURRENTLY STORED IN DATA BASE
10100   Nogo=0
10105   Zo=50                  !FARANT'S REF Zo IS ASSIGNED ONLY HERE
10110   DEG                     !DEFAULT FOR TRIG FUNCTIONS IS DEGREES
10115! USER DESCRIBES HIS CKT AND REQUESTS ANALYSIS AND OUTPUT NEXT . . .
10120 !
10125   Lcoil=X(1)*X(1)
10130   Rcoil=X(2)*X(2)
10135   Ccoil=X(3)*X(3)
10140   IF Opt=0 OR Opt=1 THEN GOTO Fval
10145 !
10150 ! FREQUENCY BAND AND STEP
10155 !
10160   Fmin=1
10165   Fmax=4
10170   Fstep=.1
10175   M=INT(DROUND((Fmax-Fmin)/Fstep,12))+1! NUMBER OF FREQUENCIES
10180   Newfile(Filedata(*),"COIL ",Fmin,Fmax,Fstep)! LOADS Filedata(*)
10185 !
10190 Fval:Fvalue=0
10195 !
10200   FOR I=1 TO M! FREQUENCY LOOP
10205     F=Fmin+(I-1)*Fstep
10210     MAT A= Filedata(I,*,*)! LOADS A(*) WITH DATA AT EACH FREQUENCY
10215     Rlc(B(*),"S",Rcoil,Lcoil,0,"P",0)
10220     Rlc(C(*),"P",0,0,Ccoil,"P",0)
10225     Cas(B(*),C(*))
10230     Rlc(C(*),"S",1.E+9,0,0,"S",0)
10235     Cas(B(*),C(*))
10240     IF Opt=2 THEN
10245       Saveckt(A(*),-1,4,0)
10250     ELSE
10255       Saveckt(B(*),-1,4,0)
10260     END IF
10265     Fvalue=Fvalue+(B(1,1)-A(1,1))^2+(B(1,2)-A(1,2))^2
10270   NEXT I
10275   IF Opt=1 THEN SUBEXIT
10280   IF Opt=2 THEN
10285     PRINT "INITIAL VALUES OF VARIABLES:"
10290   ELSE
10295     PRINT "FINAL VALUES AFTER OPTIMITATION:"
```

```
10300 END IF
10305 PRINT "LCOIL=";DROUND(Lcoil,4);"RCOIL=";DROUND(Rcoil,4);"CCOIL=";DROUND(
Ccoil,4)
10310 IF Opt=2 THEN
10315 PRINT "DATA FROM FILE:"
10320 ELSE
10325 PRINT "EQUIVALENT CIRCUIT RESULTS:"
10330 END IF
10335 Prt(4,-4)
10340 PAUSE
10345 Smith(-1,1,-1,1)
10350 Splot(1,1)
10355 PAUSE
10360 GCLEAR
10365 CONTROL 1,12;0
10370 SUBEND
```